

New Orbits for the n -Body Problem

ROBERT J. VANDERBEI

*Operations Research and Financial Engineering,
Princeton University, Princeton, New Jersey, USA*

ABSTRACT: In this paper, we consider minimizing the action functional as a method for numerically discovering periodic solutions to the n -body problem. With this method, we can find a large number of choreographies and other more general solutions. We show that most of the solutions found, including all but one of the choreographies, are unstable. It appears to be much easier to find unstable solutions to the n -body problem than stable ones. Simpler solutions are more likely to be stable than exotic ones.

KEYWORDS: new orbits; n -body problem; action principle

LEAST ACTION PRINCIPLE

Given n bodies, let m_j denote the mass and $z_j(t)$ denote the position in $\mathbb{R}^2 = \mathbb{C}$ of body j at time t . The *action functional* is a mapping from the space of all trajectories, $z_1(t), z_2(t), \dots, z_n(t)$, $0 \leq t \leq 2\pi$, into the reals. It is defined as the integral over one period of the kinetic minus the potential energy,

$$A = \int_0^{2\pi} \left(\sum_j \frac{m_j}{2} \|\dot{z}_j\|^2 + \sum_{j, k: k < j} \frac{m_j m_k}{\|z_j - z_k\|} \right) dt.$$

Stationary points of the action function are trajectories that satisfy the equations of motion; that is, Newton's laws of gravity. To see this, we compute the first variation of the action functional,

$$\begin{aligned} \delta A &= \int_0^{2\pi} \sum_{\alpha} \left(\sum_j m_j \dot{z}_j^{\alpha} \delta z_j^{\alpha} - \sum_{j, k: k < j} m_j m_k \frac{(z_j^{\alpha} - z_k^{\alpha})(\delta z_j^{\alpha} - \delta z_k^{\alpha})}{\|z_j - z_k\|^3} \right) dt \\ &= - \int_0^{2\pi} \sum_j \sum_{\alpha} \left(\sum_j m_j \dot{z}_j^{\alpha} + \sum_{k: k \neq j} m_j m_k \frac{z_j^{\alpha} - z_k^{\alpha}}{\|z_j - z_k\|^3} \right) \delta z_j^{\alpha} dt, \end{aligned}$$

and set it to zero. We obtain,

$$m_j \dot{z}_j^{\alpha} = \sum_{k: k \neq j} m_j m_k \frac{z_j^{\alpha} - z_k^{\alpha}}{\|z_j - z_k\|^3}, \quad j = 1, 2, \dots, n, \quad \alpha = 1, 2. \quad (1)$$

Note that if $m_j = 0$ for some j , then the first order optimality condition reduces to $0 = 0$, which is not the equation of motion for a massless body. Hence, we must assume that all bodies have positive mass.

Address for correspondence: Robert J. Vanderbei, Operations Research and Financial Engineering, Princeton University, Princeton, New Jersey, USA.
rvdb@princeton.edu www.princeton.edu/~rvdb

PERIODIC SOLUTIONS

Our goal is to use numeric optimization to minimize the action functional and thereby find periodic solutions to the n -body problem. Since we are interested only in periodic solutions, we express all trajectories in terms of their Fourier series,

$$z_j(t) = \sum_{k=-\infty}^{\infty} \gamma_k e^{ikt}, \quad \gamma_k \in \mathbb{C}.$$

Abandoning the efficiency of complex-variable notation, we can write the trajectories with components $z_j(t) = (x_j(t), y_j(t))$ and $\gamma_k = (\alpha_k, \beta_k)$. So doing, we get

$$\begin{aligned} x(t) &= a_0 + \sum_{k=1}^{\infty} (a_k^c \cos(kt) + a_k^s \sin(kt)) \\ y(t) &= b_0 + \sum_{k=1}^{\infty} (b_k^c \cos(kt) + b_k^s \sin(kt)), \end{aligned}$$

where

$$\begin{aligned} a_0 &= \alpha_0, & a_k^c &= \alpha_k + \alpha_{-k}, & a_k^s &= \beta_{-k} - \beta_k, \\ b_0 &= \beta_0, & b_k^c &= \beta_k + \beta_{-k}, & b_k^s &= \alpha_k - \alpha_{-k}. \end{aligned}$$

Since we plan to optimize over the space of trajectories, the parameters a_0 , a_k^c , a_k^s , b_0 , b_k^c , and b_k^s are the decision variables in our optimization model. The objective is to minimize the action functional.

AMPL is a small programming language designed for the efficient expression of optimization problems.¹ FIGURE 1 shows the AMPL program for minimizing the action functional.

Note that the action functional is a nonconvex nonlinear functional. Hence, it is expected to have many local extrema and saddle points. We use the author's local optimization software called LOQO^{2,3} to find local minima in a neighborhood of an arbitrary given starting trajectory. One can provide either specific initial trajectories or one can give random initial trajectories. The four lines just before the call to `solve` in FIGURE 1 show how to specify a random initial trajectory. Of course, AMPL provides capabilities of printing answers in any format, either on a standard output device or to a file. For the sake of brevity and clarity, the print statements are not shown in FIGURE 1. AMPL also provides the capability to loop over sections of code. This is also not shown, but the program we used has a loop around the four initialization statements, the call to solve the problem, and the associated print statements. In this way, the program can be run once to solve for a large number of periodic solutions.

Choreographies

Recently, Chenciner and Montgomery⁴ introduced a new family of solutions to the n -body problem called choreographies. A *choreography* is defined as a solution to the n -body problem in which all of the bodies share a common orbit and are uniformly spread out around this orbit. Such trajectories are even easier to find using the action principle. Rather than having a Fourier series for each orbit, it is only

```

param N := 3; # number of masses
param n := 15; # number of terms in Fourier series representation
param m := 100; # number of terms in numerical approx to integral

set Bodies := {0..N-1};
set Times := {0..m-1} circular; # "circular" means that next(m-1) = 0

param theta {t in Times} := t*2*pi/m;
param dt := 2*pi/m;

param a0 {i in Bodies} default 0;      param b0 {i in Bodies} default 0;
var as {i in Bodies, k in 1..n} := 0;   var bs {i in Bodies, k in 1..n} := 0;
var ac {i in Bodies, k in 1..n} := 0;   var bc {i in Bodies, k in 1..n} := 0;

var x {i in Bodies, t in Times}
  = a0[i]+sum {k in 1..n} ( as[i,k]*sin(k*theta[t]) + ac[i,k]*cos(k*theta[t]) );
var y {i in Bodies, t in Times}
  = b0[i]+sum {k in 1..n} ( bs[i,k]*sin(k*theta[t]) + bc[i,k]*cos(k*theta[t]) );

var xdot {i in Bodies, t in Times} = (x[i,next(t)]-x[i,t])/dt;
var ydot {i in Bodies, t in Times} = (y[i,next(t)]-y[i,t])/dt;

var K {t in Times} = 0.5*sum {i in Bodies} (xdot[i,t]^2 + ydot[i,t]^2);

var P {t in Times}
  = - sum {i in Bodies, ii in Bodies: ii>i}
    1/sqrt((x[i,t]-x[ii,t])^2 + (y[i,t]-y[ii,t])^2);

minimize A: sum {t in Times} (K[t] - P[t])*dt;

let {i in Bodies, k in 1..n} as[i,k] := 1*(Uniform01()-0.5);
let {i in Bodies, k in 1..n} ac[i,k] := 1*(Uniform01()-0.5);
let {i in Bodies, k in 1..n} bs[i,k] := 0.01*(Uniform01()-0.5);
let {i in Bodies, k in 1..n} bc[i,k] := 0.01*(Uniform01()-0.5);

solve;

```

FIGURE 1. AMPL program for finding trajectories that minimize the action functional.

necessary to have one master Fourier series and to write the action functional in terms of it. FIGURE 2 shows the AMPL model for finding choreographies.

STABLE VERSUS UNSTABLE SOLUTIONS

FIGURE 3 shows some simple choreographies found by minimizing the action functional using the AMPL model in FIGURE 2. The famous three-body figure eight, first discovered by Moore⁵ and later analyzed by Chenciner and Montgomery,⁴ is the first one shown—labeled FigureEight3. It is easy to find choreographies of arbitrary complexity. In fact, it is not hard to rediscover most of the choreographies given by Chenciner *et al.*,⁶ and more, simply by putting a loop in the AMPL model and finding various local minima by using different starting points.

However, as we discuss in a later section, simulation makes it apparent that, with the sole exception of FigureEight3, all of the choreographies we found are unstable. Furthermore, the more intricate the choreography, the more unstable it is. Since the only choreographies that have a chance to occur in the real world are stable ones, many cpu hours were devoted to searching for other stable choreographies. So far, none have been found. The choreographies shown in FIGURE 3 represent the choreographies closest to being stable.

```

param N := 3; # number of masses
param n := 15; # number of terms in Fourier series representation
param m := 99; # terms in num approx to integral. must be a multiple of N

param lagTime := m/N;

set Bodies := {0..N-1};
set Times := {0..m-1} circular; # "circular" means that next(m-1) = 0

param theta {t in Times} := t*2*pi/m;
param dt := 2*pi/m;

param a0 default 0;      param b0 default 0;
var as {k in 1..n} := 0;  var bs {k in 1..n} := 0;
var ac {k in 1..n} := 0;  var bc {k in 1..n} := 0;

var x {i in Bodies, t in Times}
  = a0+sum {k in 1..n} ( as[k]*sin(k*theta[(t+i*lagTime) mod m])
    + ac[k]*cos(k*theta[(t+i*lagTime) mod m]) );
var y {i in Bodies, t in Times}
  = b0+sum {k in 1..n} ( bs[k]*sin(k*theta[(t+i*lagTime) mod m])
    + bc[k]*cos(k*theta[(t+i*lagTime) mod m]) );

var xdot {i in Bodies, t in Times} = (x[i,next(t)]-x[i,t])/dt;
var ydot {i in Bodies, t in Times} = (y[i,next(t)]-y[i,t])/dt;

var K {t in Times} = 0.5*sum {i in Bodies} (xdot[i,t]^2 + ydot[i,t]^2);

var P {t in Times}
  = - sum {i in Bodies, ii in Bodies: ii>i}
    1/sqrt((x[i,t]-x[ii,t])^2 + (y[i,t]-y[ii,t])^2);

minimize A: sum {t in Times} (K[t] - P[t])*dt;

let {k in 1..n} as[k] := 1*(Uniform01()-0.5);
let {k in 1..n} ac[k] := 1*(Uniform01()-0.5);
let {k in n..n} bs[k] := 0.01*(Uniform01()-0.5);
let {k in n..n} bc[k] := 0.01*(Uniform01()-0.5);

solve;

```

FIGURE 2. AMPL program for finding choreographies by minimizing the action functional.




FIGURE 3. Periodic orbits—choreographies.

Given the difficulty of finding stable choreographies, it seems interesting to search for stable nonchoreographic solutions using, for example, the AMPL model from FIGURE 1. The most interesting among such solutions are shown in FIGURE 4. That labeled *Ducati3* is stable as are *Hill3_15* and the three *DoubleDouble* solutions. However, the more exotic solutions (*OrthQuasiEllipse4*, *Rosette4*, *PlateSaucer4*, and *BorderCollie4*) are all unstable.

For the interested reader, a Java applet can be found at Reference 7 that allows one to watch the dynamics of each of the systems presented in this paper (and others). This applet actually integrates the equations of motion. If the orbit is unstable it becomes very obvious as the bodies deviate from their predicted paths.

Ducati3 and Its Relatives

The *Ducati3* orbit first appeared in a paper by Moore⁵ and has been independently rediscovered by this author,⁶ and perhaps others. Simulation reveals it to be a stable system. The Java applet⁷ allows one to rotate the reference frame as desired. By setting the rotation to counter the outer body in *Ducati3*, one discovers that the other two bodies are orbiting each other in nearly circular orbits. In other words, the first body in *Ducati3* is executing an approximately circular orbit, $z_1(t) = -e^{it}$, the second body is oscillating back and forth, roughly along the x -axis, $z_2(t) = \cos(t)$, and the

third body is oscillating up and down the y -axis, $z_3(t) = i \sin(t)$. Rotating so as to fix the first body means multiplying by e^{it}

$$\bar{z}_1(t) = e^{-it}(-e^{it}) = -1$$

$$\bar{z}_2(t) = e^{-it} \cos(t) = \frac{1}{2}(1 + e^{-2it})$$

$$\bar{z}_3(t) = e^{-it} i \sin(t) = \frac{1}{2}(1 - e^{-2it}).$$

Now it is clear that bodies 2 and 3 are orbiting each other at half the distance of body 1. Thus, this system can be described as a sun, earth, moon system in which all three bodies have equal mass and in which one (sidereal) month equals one year. The synodic month is shorter—half a year.




FIGURE 4. Periodic orbits—non-choreographies.

This analysis of Ducati3 suggests looking for other stable solutions of the same type but with different resonances between the length of a month and a year. Hill3_15 is one of many such examples we found. In Hill3_15, there are 15 sidereal months per year. Let Hill3_ n denote the system in which there are n months in a year. All of these orbits are easy to calculate and they all appear to be stable. This success suggests going in the other direction. Let Hill3_1/ n denote the system in which there are n years per month. We computed Hill3_1/2 and found it to be unstable. It is shown in FIGURE 5.

In the preceding discussion, we decomposed these Hill-type systems into two two-body problems: the earth and moon orbit each other while their center of mass orbits the sun. This suggests that we can find stable orbits for the four-body problem by splitting the sun into a binary star. This works. The orbits labeled DoubleDoublen are of this type. As already mentioned, these orbits are stable.

Given the existence and stability of FigureEight3, one often is asked if there is any chance to observe such a system among the stars. The answer is that it is very unlikely since its existence depends crucially on the masses being equal. The Ducati and Hill type orbits, however, are not constrained to have their masses be equal. FIGURE 6 shows several Ducati-type orbits in which the masses are not all equal. All of these orbits are stable. This suggests that stability is common for Ducati and Hill type orbits. Perhaps such orbits can be observed.

LIMITATIONS OF THE MODEL

There are certain limitations to the approach articulated above. First, the Fourier series is an infinite sum that gets truncated to a finite sum in the computer model. Hence, the trajectory space from which solutions are found is finite dimensional.

Second, the integration is replaced with a Riemann sum. If the discretization is too coarse, the solution found might not correspond to a real solution to the n -body problem. The only way to be sure is to run a simulator.

Third, as mentioned previously, all masses must be positive. If there is a zero mass, then the stationary points for the action function, which satisfy (1), do not necessarily satisfy the equations of motion given by Newton's laws.

Last, the model, as given in FIGURE 1, cannot solve two-body problems with eccentricity. We address this issue in the next section.

ELLIPTIC SOLUTIONS

An ellipse with semimajor axis a , semiminor axis b , and having its left focus at the origin of the coordinate system is given parametrically by

$$x(t) = f + a \cos t, \quad y(t) = b \sin t,$$

where $f = \sqrt{a^2 - b^2}$ is the distance from the focus to the center of the ellipse.

However, this is *not* the trajectory of a mass in the two-body problem. Such a mass will travel faster around one focus than around the other. To accommodate this, we need to introduce a time-change function $\theta(t)$,

$$x(t) = f + a \cos \theta(t), \quad y(t) = b \sin \theta(t).$$




FIGURE 5. Periodic orbits—Ducati’s with unequal masses.




FIGURE 6. Periodic orbits—Hill-type with equal masses.

This function θ must be increasing and must satisfy $\theta(0) = 0$ and $\theta(2\pi) = 2\pi$.

The optimization model can be used to find (a discretization of) $\theta(t)$ automatically by changing `param theta` to `var theta` and adding appropriate monotonicity and boundary constraints. In this manner, more realistic orbits can be found that could be useful in real space missions.

In particular, using an eccentricity $e = f/a = 0.0167$ and appropriate sun and earth masses, we can find a periodic Hill-type satellite trajectory in which the satellite orbits the earth once per year.

SENSITIVITY ANALYSIS

The determination of stability versus instability mentioned the previous sections was done empirically, by simulating the orbits with an integrator and very small step sizes. Two integrators were used: a midpoint integrator and a fourth order Runge–Kutta integrator. Orbits that are claimed to be stable were run for several hours of cpu time (which corresponds to many thousands of orbits) without falling apart. Orbits that are claimed to be unstable generally became obviously so in just a few seconds of cpu time, which corresponds to only a few full orbits. In this section, we describe a Floquet analysis of stability and present this measure of stability for the various orbits found.

For simplicity, in this section we assume that all masses are equal to one. Let $\xi^*(t) = (z^*(t), \dot{z}^*(t))$ be a particular solution to

$$\dot{\xi} = A(\xi),$$

where

$$A(z(t), \dot{z}(t)) = (\dot{z}(t), a(z(t))),$$

and

$$a(z) = (a_1(z), \dots, a_n(z)),$$

and

$$a_j(z) = \sum_{k: k \neq j} \frac{z_j - z_k}{\|z_j - z_k\|^2}, \quad j = 1, 2, \dots, n.$$

Consider a nearby solution $\xi(t)$,

$$\begin{aligned} \dot{\xi}(t) &= A(\xi(t)) \\ &\approx A(\xi^*(t)) + A'(\xi^*(t))(\xi(t) - \xi^*(t)) \\ &= \dot{\xi}^*(t) + A'(\xi^*(t))(\xi(t) - \xi^*(t)). \end{aligned}$$

Put $\Delta\xi = \xi - \xi^*$. Then $\dot{\Delta\xi} = A'(\xi^*(t))\Delta\xi$. A finite difference approximation yields

$$\begin{aligned} \Delta\xi(t+h) &= \Delta\xi(t) + hA'(\xi^*(t))\Delta\xi(t) \\ &= (I + hA'(\xi^*(t)))\Delta\xi(t). \end{aligned}$$

Iterating around one period, we obtain

$$\Delta\xi(T) = \left(\prod_{i=0}^{n-1} (I + hA'(\xi^*(t_i))) \right) \Delta\xi(0),$$

where $h = T/n$ and $t_i = iT/n$.

The following perturbations, which are associated with invariants of the physical laws, are unimportant in calculating $\Delta\xi(T)$:

$$\begin{bmatrix} \Delta z \\ \Delta \dot{z} \end{bmatrix} = \begin{bmatrix} e_1 \\ e_1 \\ e_1 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} e_2 \\ e_2 \\ e_2 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 0 \\ e_1 \\ e_1 \\ e_1 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 0 \\ e_2 \\ e_2 \\ e_2 \end{bmatrix}, \begin{bmatrix} Rz_1 \\ Rz_2 \\ Rz_3 \\ R\dot{z}_1 \\ R\dot{z}_2 \\ R\dot{z}_3 \end{bmatrix}, \frac{1}{2} \begin{bmatrix} -3\dot{z}_1 + 2z_1 \\ -3\dot{z}_2 + 2z_2 \\ -3\dot{z}_3 + 2z_3 \\ -3a_1 - \dot{z}_1 \\ -3a_2 - \dot{z}_2 \\ -3a_3 - \dot{z}_3 \end{bmatrix},$$

where R denotes rotation by 90° . The first two of these perturbations correspond to *translation*. The next two correspond to a *moving frame of reference* and the last two correspond to *rotation* and *dilation*. Dilation is explained below. Of course, all positions and velocities are evaluated at $t = 0$. Vector e_i denotes the i th unit vector in \mathbb{R}^2 .

Consider spatial dilation by ρ together with a temporal dilation by θ ,

$$Z_j(t) = \rho z_j(t/\theta).$$

Given that the z_j values are a solution, it is easy to check that

$$\ddot{Z}_j(t) = -\frac{\rho^3}{\theta^2} \sum_{k \neq j} \frac{Z_j(t) - Z_k(t)}{\|Z_j(t) - Z_k(t)\|^2}.$$

Hence, if mass is to remain fixed, we must have that $\rho^3 = \theta^2$,

$$Z_j(t) = \rho z_j(t/\rho^{3/2}), \quad \dot{Z}_j(t) = \rho^{-1/2} \dot{z}_j(t/\rho^{3/2}).$$

To find the perturbation direction corresponding to this dilation, we differentiate with respect to ρ at $\rho = 1$,

$$\frac{d}{d\rho} \begin{bmatrix} \rho z_j(t/\rho^{3/2}) \\ \rho^{-1/2} \dot{z}_j(t/\rho^{3/2}) \end{bmatrix} \bigg|_{\rho=1} = \begin{bmatrix} -\frac{3}{2} \dot{z}_j + z_j \\ -\frac{3}{2} a_j - \frac{1}{2} \dot{z}_j \end{bmatrix}.$$

To check stability, we project any initial perturbation onto the null space of P^T , where

$$P = \begin{bmatrix} e_1 & e_2 & 0 & 0 & Rz_1 & (-3\dot{z}_1 + 2z_1)/2 \\ e_1 & e_2 & 0 & 0 & Rz_2 & (-3\dot{z}_2 + 2z_2)/2 \\ e_1 & e_2 & 0 & 0 & Rz_3 & (-3\dot{z}_3 + 2z_3)/2 \\ 0 & 0 & e_1 & e_2 & R\dot{z}_1 & (-3a_1 + \dot{z}_1)/2 \\ 0 & 0 & e_1 & e_2 & R\dot{z}_2 & (-3a_2 + \dot{z}_2)/2 \\ 0 & 0 & e_1 & e_2 & R\dot{z}_3 & (-3a_3 + \dot{z}_3)/2 \end{bmatrix}.$$

The projection matrix is given by

$$\Pi = I - P(P^T P)^{-1} P^T.$$

TABLE 1. Apparently stable orbits

Name	$\max(\lambda_i(\Lambda))$	$\max(\lambda_i(\Lambda\Pi))$
Lagrange2	1.383	1.362
FigureEight3	1.228	4.220
Ducati3	1.105	3.885
Hill3_15	1.444	2.403
DoubleDouble5	12.298	12.298
DoubleDouble10	1.404	5.948
DoubleDouble20	1.890	1.890

From the fact that $z_1 + z_2 + z_3 = 0$ and $\dot{z}_1 + \dot{z}_2 + \dot{z}_3 = 0$, it follows that all columns of P are mutually orthogonal *except* for the fifth and sixth columns. Hence, $P^T P$ is not a purely diagonal matrix.

Let

$$\Lambda_n = \left(\prod_{i=0}^{n-1} (I + hA'(\xi^*(t_i))) \right).$$

We say that an orbit is *stable* if all eigenvalues of

$$\lim_{n \rightarrow \infty} \Lambda_n \Pi$$

are at most one in magnitude.

TABLE 2. Apparently unstable orbits

Name	$\max(\lambda_i(\Lambda))$	$\max(\lambda_i(\Lambda\Pi))$
Lagrange3	81.630	81.630
OrthQuasiEllipse4	18.343	18.343
Rosette4	1.873	4.449
Braid4	727.508	711.811
Trefoil4	41,228.515	41,213.852
FigureEight4	221.642	194.095
FoldedTriLoop4	74,758.355	74,675.092
PlateSaucer4	3,653.210	3,653.210
BorderCollie4	188.235	188.052
Trefoil5	1.913×10^8	1.917×10^8
FigureEight5	2,223.137	2,223.457

Stable Orbits

We computed Λ_n for $n = 10^6$. TABLE 1 shows maximum eigenvalues for those orbits that seemed stable from simulation. TABLE 2 shows maximum eigenvalues for those orbits that appeared unstable when simulated.

ACKNOWLEDGMENTS

The author received support from the NSF (CCR-0098040) and the ONR (N00014-98-1-0036).

Notes Added in Proof

1. The integration method referred to as the midpoint integrator in the section SENSITIVITY ANALYSIS is the same as the leapfrog method of Hut *et al.*^{8,9}
2. Roger Broucke has informed me that Ducati3 and the Hill type solutions first appeared in his papers,^{10–12} where their stability was noted.

REFERENCES

1. FOURER, R., D.M. GAY & B.W. KERNIGHAN. 1993. AMPL: A Modeling Language for Mathematical Programming. Scientific Press.
2. VANDERBEI, R.J. 1999. LOQO user's manual—version 3.10. Optimiz. Meth. Software **12**: 485–514.
3. VANDERBEI, R.J. & D.F. SHANNO. 1999. An interior-point algorithm for nonconvex nonlinear programming. Comput. Optimiz. Appl. **13**: 231–252.
4. CHENCINER, A. & R. MONTGOMERY. 2000. A remarkable periodic solution of the three-body problem in the case of equal masses. Ann. Math. **152**: 881–901.
5. MOORE, C. 1993. Braids in classical gravity. Phys. Rev. Lett. **70**: 3675–3679.
6. CHENCINER, A., J. GERVER, R. MONTGOMERY & C. SIMÓ. 2002. Simple choreographic motions on n bodies: a preliminary study. In Geometry, Mechanics and Dynamics. Springer, New York. 287–308.
7. VANDERBEI, R.J. 2001. <<http://www.princeton.edu/~rvdb/JAVA/astro/galaxy/Galaxy.html>>.
8. HUT, P. & D. HEGGIE. 2003. The Gravitational Millron-Body Problem. Cambridge University Press.
9. HUT, P., J. MAKINO & S. McMILLAN. 1995. Making a better leapfrog. ApJ. 443L, 93.
10. BROUCKE, R. & D. BOGGS. 1975. Periodic orbits in the planar general three-body problem. Celest. Mechan. **11**(1): 13–38.
11. BROUCKE, R. 1975. On relative periodic solutions of the planar general three-body problem. Celest. Mechan. **12**(4): 439–462.
12. BROUCKE, R. 1969. Stability of periodic orbits. AIAA J. **7**(6): 1003–1009.